SIF5005 våren 2003: Maple-øving 1

Løsningsforslag.

1 Oppgave 1. Litt grunnleggende Maple

Hvordan får du hjelp i Maple med en funksjon når du kjenner navnet? Det raskeste er slik:

```
> ?simplify
```

Tips for å lese hjelpesider: Det er ofte kjekt å ta en rask kikk på beskrivelsen øverst, og så gå rett til eksemplene nederst. Gå tilbake til beskrivelsen om det er noen detaljer som er uklare.

Hva gjør kommandoene ***sum***, ***limit***, ***int***? Og hva med ***Sum***, ***Limit***, ***Int***? Hva med ***value***?

Svar: sum, limit, int prøver å regne ut lukkede uttrykk for en sum, en grense, eller et integral. De returneres uevaluert om det ikke går. De tilsvarende med stor forbokstav returneres alltid uevaluert. Anvend value på dem for å bytte ut Sum med sum, etc., og så evaluere resultatet.

Her er noen tilfeldig utvalgte eksempler:

```
> Int(exp(-t^2),t=0..x);
```

$$\int_0^x e^{(-t^2)}\,dt$$

```
> value(%);
```

$$\frac{1}{2}\sqrt{\pi}\,\mathrm{erf}(x)$$

```
> sum(i^4,i=1..n);
```

$$\frac{1}{5}(n+1)^5-\frac{1}{2}(n+1)^4+\frac{1}{3}(n+1)^3-\frac{1}{30}n-\frac{1}{30}$$

```
> sum(i^4,i=1..10);
```

$$25333$$

```
> sum(f(i),i=1..10);
```

$$\mathrm{f}(1)+\mathrm{f}(2)+\mathrm{f}(3)+\mathrm{f}(4)+\mathrm{f}(5)+\mathrm{f}(6)+\mathrm{f}(7)+\mathrm{f}(8)+\mathrm{f}(9)+\mathrm{f}(10)$$

```
> sum(f(i),i=1..infinity);
```

$$\sum_{i=1}^{\infty}\mathrm{f}(i)$$

```
> Sum(i^3,i=1..n);
```

$$\sum_{i=1}^{n} i^3$$

```
> %=value(%);
```

$$\sum_{i=1}^{n} i^3=\frac{1}{4}(n+1)^4-\frac{1}{2}(n+1)^3+\frac{1}{4}(n+1)^2$$

Hva er forskjellen på en **funksjon** og et **uttrykk** i Maple? Sammenlign operatorene ***D*** og ***diff***.

(Legg merke til at Maple ikke gjør forskjell på ordinær derivasjon og partiell derivasjon, i hvert fall ikke når du deriverer *uttrykk*.)

Svar: En **funksjon** er, som i matematikken ellers, en *regel* som tar en eller flere parametre som input og returnerer en verdi. Her definerer vi for eksempel en funksjon f av to variable, og regner deretter ut noen verdier - både numeriske og symbolske:

```
> f:=(x,y)->x^2-y^2;
```

$$f := (x,\,y) \to x^2-y^2$$

```
> f(2,3);
```

$$-5$$

```
> f(a,b);
```

$$a^2-b^2$$

```
> f(x+y,x-y);
```

$$(x+y)^2 - (x-y)^2$$

```
> simplify(%);
```

$$4\,x\,y$$

Et **uttrykk** derimot, er bare en formel. Selv om vi i dagligtalen kan snakke om «funksjonen $x^2 - y^2$», er det egentlig nonsens å snakke slik.

```
> g:=x^2-y^2;
```

$$g := x^2 - y^2$$

Vi får nokså meningsløse resultater om vi forsøket å anvende dette utrykket som om det var en funksjon:

```
> g(2,3);
```

$$\mathrm{x}(2,\,3)^2 - \mathrm{y}(2,\,3)^2$$

Forklaring: Dette ser meningsløst ut, men gir likevel mening om du tenker deg at x og y er (ukjente/udefinerte) funksjoner av to variable: Da kan uttrykket $x^2 - y^2$ oppfattes som en funksjon av to variable og. Men skal du bruke g som om det var en funksjon, trenger du å bruke ***subs*** for å substituere variable i g:

```
> subs(x=2,y=3,g);
```

$$-5$$

```
> subs(x=a,y=b,g);
```

$$a^2 - b^2$$

```
> subs(x=x+y,y=x-y,g);
```

$$(2\,x - y)^2 - (x-y)^2$$

Det var ikke akkurat hva vi hadde tenkt oss! Det ovenstående er ekvivalent med å gjøre disse to substitusjonene etter hverandre:

```
> subs(x=x+y,g);
> subs(y=x-y,%);
```

$$(x+y)^2 - y^2$$

$$(2\,x - y)^2 - (x-y)^2$$

Problemet er at den første substitusjonen innfører en ny y som så blir substituert i den andre. Problemet kan unngås slik:

```
> subs({x=x+y,y=x-y},g);
```

$$(x+y)^2 - (x-y)^2$$

Mengdeparentesen indikerer at rekkefølgen skal være uviktig, og får Maple til å gjøre substitusjonene «i parallell».

Men enklere er det nok likevel å jobbe med funksjoner, det meste av tiden. Ofte får man et uttrykk som resultat av en beregning, og ønsker å lage en funskjon av det. Da er ***unapply*** hendig:

```
> f:=unapply(g,x,y);
```

$$f := (x,\,y) \to x^2 - y^2$$

D er til for å derivere *funksjoner*, ***diff*** for å derivere *uttrykk*.

```
> diff(g,x);
```

$$2\,x$$

D alene deriverer bare funksjoner av en variabel, så dette går galt:

```
> D(f);

Error, (in D/procedure) function must be unary
```

Vi trenger en notasjon som vi ellers ikke ser så ofte, men den er beskrevet i E&P:

```
> D[1];
```

$$\mathrm{D}_1$$

```
> D[1](f);
```

$$(x,\,y) \to 2\,x$$

D og ***diff*** vil gjerne derivere *ukjente* funksjoner:

```
> D(h)(x)=diff(h(x),x);
```

$$\mathrm{D}(h)(x) = \frac{\partial}{\partial x}\,\mathrm{h}(x)$$

Vi fikk svaret uttrykt som om det var partiellderivasjon vi ba om, enda det bare er en variabel i uttrykket. Siden kan vi godt substituere inn en kjent funksjon for h, så får vi «skikkelige» verdier:

```
> subs(h=sin,%);
```

$$\mathrm{D}(\sin)(x) = \frac{\partial}{\partial x}\sin(x)$$

Men ***subs*** er litt lur: Maple gjør bare minimale forenklinger etter en substitusjon, så vi må tvinge den til å evaluere resultatet med ***eval***:

```
> eval(%);
```

$$\cos(x) = \cos(x)$$

2 Oppgave 2. Plotting av en funksjon, dens deriverte og antideriverte

```
> f:=x->exp(x)*sin(x^2);
```

$$f := x \to e^x \sin(x^2)$$

```
> int(f(x),x);
```

$$-\frac{\frac{1}{4}\,I\,\sqrt{\pi}\,e^{(1/4\,I)}\,\mathrm{erf}(\sqrt{-I}\,x - \frac{1}{2}\,\frac{1}{\sqrt{-I}})}{\sqrt{-I}} + \frac{1}{4}\,(-1)^{(1/4)}\,\sqrt{\pi}\,e^{(-1/4\,I)}\,\mathrm{erf}((-1)^{(1/4)}\,x + \frac{1}{2}\,(-1)^{(3/4)})$$

Dette svaret inneholder komplekse tall, og er neppe så nyttig for plotting. Bedre å jobbe med integralet **uevaluert**, kanskje?

```
> plot(Int(f(t),t=0..x),x=-2..5);
```



Ja, det virket. Alle sammen i ett plott:

```
> plots[display](
> plot(f(x),x=-2..5,color=red),
> plot(D(f)(x),x=-2..5,color=green),
> plot(Int(f(t),t=0..x),x=-2..5,color=blue));
```



Det var kanskje ikke så veldig instruktivt? Den deriverte dominerer totalt, og integralet er knapt synlig. Etter litt eksperimentering endte jeg med å multiplisere med disse konstantene, for å få alle grafene omtrent like store:

```
>  plots[display](
>  plot(0.1*f(t),t=-2..5,color=red),
>  plot(0.01*D(f)(t),t=-2..5,color=green),
>  plot(Int(f(t),t=0..x),x=-2..5,color=blue));
```



3 Oppgave 3. «Kreyszig-nøtta»

Her er differensialligningen som var oppgitt:

```
> difflign:=(x^2 - 2*x)*diff(y(x),x) = 2*(x-1)*y(x):  difflign;
```

$$(x^2 - 2\,x)\,(\tfrac{\partial}{\partial x}\,\mathrm{y}(x)) = 2\,(x-1)\,\mathrm{y}(x)$$

Vi løser den med initialbetingelse $\mathrm{y}(x_0) = y_0$:

```
> dsolve({difflign, y(x0) = y0});
```

$$\mathrm{y}(x) = \frac{y0\,x\,(x-2)}{x0\,(x0-2)}$$

Åpenbart oppstår et problem dersom vi får null i nevner, altså for $x_0 = 0$ eller $x_0 = 2$. En tegning av retningsfeltet ser ut til å fortelle samme historie:

```
> DEtools[DEplot](difflign, y(x), x=-1..3, y=-3..3);
```



Vi kan utforske litt mer forsiktig hva som skjer i de to «vanskelige» tilfellene ved å finne den *generelle* løsningen av ligningen:

```
> dsolve(difflign); genlosn:=%:
```

$$\mathrm{y}(x) = \_C1\,x\,(x-2)$$

```
> subs(y(x)=y0,x=x0,genlosn);
```

$$y0 = \_C1\,x0\,(x0-2)$$

Dette viser klart at så lenge x_0 ikke er 0 eller 2, kan konstanten velges entydig så initialbetingelsen er oppfylt, og vi har en entydig løsning.

Hvis $x_0 = 0$ eller $x_0 = 2$ har vi to muligheter: Om $y_0 \neq 0$ har ligningen ingen løsning, men om $y_0 = 0$ har vi (i hvert fall tilsynelatende) uendelig mange løsninger, for alle verdier av konstanten oppfyller løsningen. Og ser vi på den generelle løsningen så er det jo også opplagt at den oppfyller $\mathrm{y}(0) = 0$ uansett hva konstanten er.

4 Oppgave 4. Vektorer

```
> with(LinearAlgebra,DotProduct,CrossProduct);
```

$$[DotProduct,\ CrossProduct]$$

```
> a:=<2,5,8>; b:=<3,4,-2>; c:=<1,-2,1>;
```

$$a := \begin{bmatrix} 2 \\ 5 \\ 8 \end{bmatrix}$$

$$b := \begin{bmatrix} 3 \\ 4 \\ -2 \end{bmatrix}$$

$$c := \begin{bmatrix} 1 \\ -2 \\ 1 \end{bmatrix}$$

Her er prikkproduktene. $a\ .\ c = 0$ viser at a og c er ortogonale med hverandre:

```
> 'a.b'=a.b, 'a.c'=a.c, 'b.c'=b.c;
```

$$a\ .\ b = 10,\ a\ .\ c = 0,\ b\ .\ c = -7$$

Og her kommer kryssproduktene. De er alle forskjellig fra null, så ingen av vektorene er parallelle med hverandre:

```
> `axb`=CrossProduct(a,b), `axc`=CrossProduct(a,c),
> `bxc`=CrossProduct(b,c);
```

$$axb = \begin{bmatrix} -42 \\ 28 \\ -7 \end{bmatrix},\ axc = \begin{bmatrix} 21 \\ 6 \\ -9 \end{bmatrix},\ bxc = \begin{bmatrix} 0 \\ -5 \\ -10 \end{bmatrix}$$

Her er et bilde av vektorene. Jeg hadde ikke med orientation-argumentet først. I stedet brukte jeg musa for å rotere på figuren inntil jeg hadde en brukbar orientering, så leste jeg vinklene ut av øvre venstre hjørne i Maplevinduet.

```
> plots[arrow]([a,b,c],axes=normal,labels=["x","y","z"],orientation=[60
> ,75]);
```



Ja, men hvilken vektor er hvilken? Jeg vil sette navn på de med ***textplot3d***. Uheldigvis vil den funksjonen ikke ha vektorer, men bare koordinater. Hvordan finner vi koordinatene? Underavsnittet her inneholder noen eksperimenter for å finne ut hvordan får til det. Hopp over om du vil.

```
> «1,2>,<4,5»;
```

$$\begin{bmatrix} 1 \\ 2 \\ 4 \\ 5 \end{bmatrix}$$

4.1 Eksperiment

Med ***op***?

```
> op(a);
```

$3, \{(1) = 2, (2) = 5, (3) = 8\}, \mathit{datatype} = \mathit{anything}, \mathit{storage} = \mathit{rectangular}, \mathit{order} = \mathit{Fortran\_order}, \mathit{shape} = []$

Nei, det var ikke så vellykket. Vil ***convert*** hjelpe oss? Ja, vi kan i hvert fall lage en liste:

```
> convert(a,list);
```

$$[2, 5, 8]$$

Men jeg trenger å pakke ut listen også, og da er i hvert fall ***op*** nyttig likevel:

```
> op(convert(a,list));
```

$$2, 5, 8$$

```
> plots[display3d](
> plots[arrow]([a,b,c],width=[0.1,relative=false]),
> plots[textplot3d]([
> [op(convert(a,list)),"a"],
> [op(convert(b,list)),"b"],
> [op(convert(c,list)),"c"]],
> align=RIGHT,color=black),
> axes=normal,labels=["x","y","z"],orientation=[60,75]);
```



5 Oppgave 5. Flater

Implisitt plott for $x^3 + y^2 = 2\,xz$:

```
> plots[implicitplot3d](x^2+y^2=2*x*z,x=-1..1,y=-1..1,z=-1..1,grid=[20,
> 20,20],axes=boxed,orientation=[60,95]);
```



Plott av flaten $z = x^2 + y^2 + c\,x\,y$: Her med $c = 4$, som er satt rett inn i formelen (som er det enkleste, når størrelsen står bare ett sted):

```
> plot3d(x^2+y^2+4*x*y,x=-1..1,y=-1..1,
> axes=normal,style=patchcontour,orientation=[75,65]);
```



Animasjon av flaten for forskjellige verdier av c:

```
> plots[animate3d](x^2+y^2+c*x*y,x=-1..1,y=-1..1,c=-4..4,
> axes=normal,style=patchcontour,frames=20);
```

```
Error, (in plots[animate3d]) bad range arguments x = -1 .. 1, y = -1
.. 1, (Vector[column](3, [...], datatype = anything, storage =
rectangular, order = Fortran_order, shape = [])) = -4 .. 4
```

Åja, hvorfor skjedde så det? Jo, fordi vi definerte c til å være en vektor tidligere, men nå vil jeg bare bruke den som en variabel. Vi kunne bytte ut c med en annen bokstav, men det er enklere å gjøre slik:

```
> c:='c';
```

$$c := c$$

```
> plots[animate3d](x^2+y^2+c*x*y,x=-1..1,y=-1..1,c=-4..4,
> axes=normal,style=patchcontour,frames=20);
```

Enda morsommere variant av ovenstående (husk å trykke på knappen som lar animasjonen gå og gå):

```
> plots[animate3d](x^2+y^2+4*sin(t)*x*y,x=-1..1,y=-1..1,t=0..2*Pi,
> axes=normal,style=patchcontour,frames=20);
```